Design Programming

DECO1012 & DECO2011

Rob Saunders
Rob Saunders

web:  http://www.arch.usyd.edu.au/~rob

e-mail:  rob@arch.usyd.edu.au

office: Room 274, Wilkinson Building
Arrays & Pixels
Data-Structures

Data-structure is a general term used to describe different ways of organising data in a computer’s memory so that they can be accessed or manipulated easily.

Some examples of common data-structures include arrays, lists, sets, maps and trees.

We’re only going to look at the most common type: arrays
Arrays

An array is a way to store multiple items of data in memory so that they are accessed using a single variable.

Each item of data in an array is given an index that can be used to refer to it.

Arrays can be used whenever a program needs to store multiple copies of same type of data in one place.
An Array of Characters

Index 0 1 2 3 4 5 6 7 8 9
Data ↓↓↓↓↓↓↓↓↓↓↓↓
   PROCCESSING
Declaring an Array

Declaring a variable that points to an array is similar to declaring variables that point to primitive data types.

The type of the data to be stored in the array is simply followed by square brackets.

```
int[] arrayOfInts;
```
Allocating an Array

Before an array can be used the memory that it needs to store all the data has to be allocated. Memory is allocated using the “new” operator.

To allocate memory for an array the new operator needs to know the type of number of data items that will be stored.
Allocating an Array

An array is allocated by following the new operator by the type of data followed by the size of the array in square brackets:

```
arrayOfInts = new int[42];
```

It is also possible to use integer variable types to calculate the size of an array:

```
array0fInts = new int[x + 5];
```
Declaring and Allocating Arrays

It is also possible to declare and allocate an array in a single statement:

```java
int[] arrayOfInts = new int[42];
```

Unlike primitive data types however, this does not assign the values of the array.

The program has to do this later by setting the value of each data item.
Assigning Values to Arrays

To access an item we use the variable name followed by the index in square brackets.

```java
char[] vowels = new char[5];
vowels[0] = 'a';
vowels[1] = 'e';
vowels[2] = 'i';
vowels[3] = 'o';
vowels[4] = 'u';
```

Notice how the indices start at 0. This array with five items has indices 0...4.
Assigning Values to Arrays

It is possible to assign the values stored in an array don’t need to change using a special type of array allocation:

```plaintext
arrayOfPrimes = {1, 2, 3, 5, 7};
```

This allocates an array big enough to hold five integers and initialises their values to the first five prime numbers.
Declaring, Creating and Assigning Arrays

```java
int[] data; // Declare
data = new int[5]; // Create
data[0] = 19; // Assign
data[1] = 40;
data[2] = 75;
data[3] = 76;
data[4] = 90;
```

```java
int[] data = new int[5]; // Declare and create
data[0] = 19; // Assign
data[1] = 40;
data[2] = 75;
data[3] = 76;
data[4] = 90;
```

```java
// Declare, create, and assign
int[] data = { 19, 40, 75, 76, 90 };
```
Array Index Out Of Bounds Exception

- If you try to access a member of the array that lies outside the array boundaries, your program will terminate and give an `ArrayIndexOutOfBoundsException`.

```java
int[] data = { 19, 40, 75, 76, 90 };
println(data[0]); // Prints 19 to the console
println(data[2]); // Prints 75 to the console
println(data[5]); // ERROR! The last element of the array is 4
```
Array Length

- The `length` field stores the number of elements in an array.

```java
int[] data1 = { 19, 40, 75, 76, 90 };
int[] data2 = { 19, 40 };
int[] data3 = new int[127];
println(data1.length); // Prints "5" to the console
println(data2.length); // Prints "2" to the console
println(data3.length); // Prints "127" to the console
```
Using **for** Loops with Arrays

- A **for** loop can be used to access array elements, especially with large arrays.

- The following example uses a **for** structure to iterate through every value in an array.

```java
int[] data = { 19, 40, 75, 76, 90 };  
for (int i = 0; i < data.length; i++) {
    line(data[i], 0, data[i], 100);
}
```
Using **for** Loops with Arrays

float[] sineWave = new float[width];

    for (int i = 0; i < width; i++) {
        // Fill the array with values from sin()
        float r = map(i, 0, width, 0, TWO_PI);
        sineWave[i] = abs(sin(r));
    }

    for (int i = 0; i < sineWave.length; i++) {
        // Set the stroke values to numbers read from the array
        stroke(sineWave[i] * 255);
        line(i, 0, i, height);
    }
Using Arrays to Draw

- Arrays can make the task of programming much easier.
  - While it’s not necessary to use them, they can be valuable structures for managing data.
  - Arrays can be created to hold any type of data, and each element can be individually assigned and read.
Using Arrays to Draw
Using Arrays to Draw

› The star has 10 vertex points, each with 2 values, for a total of 20 data elements.
› The code on the left demonstrates using 20 separate variables.
› The code in the middle uses 10 arrays, one for each point of the shape.
› The code on the right shows the data elements can be grouped logically together in two arrays, one for the x-coordinates and one for the y-coordinates.
Separate variables

```java
int x0 = 50;
int y0 = 18;
int x1 = 61;
int y1 = 37;
int x2 = 83;
int y2 = 43;
int x3 = 69;
int y3 = 60;
int x4 = 71;
int y4 = 82;
int x5 = 50;
int y5 = 73;
int x6 = 29;
int y6 = 82;
int x7 = 31;
int y7 = 60;
int x8 = 17;
int y8 = 43;
int x9 = 39;
int y9 = 37;
```

One array for each point

```java
int[] p0 = { 50, 18 };
int[] p1 = { 61, 37 };
int[] p2 = { 83, 43 };
int[] p3 = { 69, 60 };
int[] p4 = { 71, 82 };
int[] p5 = { 50, 73 };
int[] p6 = { 29, 82 };
int[] p7 = { 31, 60 };
int[] p8 = { 17, 43 };
int[] p9 = { 39, 37 };
```

One array for each axis

```java
int[] x = { 50, 61, 83, 69, 71, 50, 29, 31, 17, 39 };
int[] y = { 18, 37, 43, 60, 82, 73, 82, 60, 43, 37 };
```
Using Arrays to Draw

```java
int[] x = {50, 61, 83, 69, 71, 50, 29, 31, 17, 39};
int[] y = { 18, 37, 43, 60, 82, 73, 82, 60, 43, 37 };

smooth();
strokeWeight(3);
beginShape();
// Reads one array element every
// time through the for loop
for (int i = 0; i < x.length; i++) {
    vertex(x[i], y[i]);
}
endShape(CLOSE);
saveFrame();
```
Using Arrays to Draw

- Storing the coordinates of many elements is another way to use arrays to make a program easier to read and manage.
  - In the following example, the `x[]` array stores the x-coordinate for each line, and the `speed[]` array stores the amount to add to `x` at each frame.
  - Using arrays allows the code to be shorter and more flexible; changing the value assigned to `numLines` sets the number of elements drawn to the screen.
int numLines = 12;
float[] x = new float[numLines];
float[] speed = new float[numLines];
float offset = 8; // Set space between lines

void setup() {
    size(100, 100);
    smooth();
    strokeWeight(10);
    for (int i = 0; i < numLines; i++) {
        x[i] = i; // Set initial position
        speed[i] = 0.1 + (i / offset); // Set initial speed
    }
}
void draw() {
    background(204);
    for (int i = 0; i < x.length; i++) {
        x[i] += speed[i];
        if (x[i] > (width + offset)) {
            x[i] = -offset * 2;
        }
        float y = i * offset;
        line(x[i], y, x[i] + offset, y + offset);
    }
}
Array Functions

› Processing provides a number of functions for manipulating arrays

› The append() function expands an array by one element, adds data to the new position, and returns the new array:

```java
String[] trees = { "ash", "oak" };
trees = append(trees, "maple"); // Add "maple" to the end
println("trees =");
println(trees); // Prints "ash oak maple"
```
Using the **append()** Function

```java
String[] trees = { "ash", "oak" };
append(trees, "maple"); // INCORRECT! Does not change the array
println("trees = ");
println(trees); // Prints "ash oak"

trees = append(trees, "maple"); // Add "maple" to the end
println("trees = ");
println(trees); // Prints "ash oak maple"

// Add "beech" to the end of the trees array, and creates a new
// array to store the change
String[] moretrees = append(trees, "beech");
println("moretrees = ");
println(moretrees); // Prints "ash oak maple beech"
```
Array Functions

- The `shorten()` function decreases an array by one element by removing the last element and returns the shortened array.

```java
String[] trees = { "lychee", "coconut", "fig"};

trees = shorten(trees); // Remove last element from the array
println("trees = ");
println(trees); // Prints "lychee coconut"

trees = shorten(trees); // Remove last element from the array
println("trees = ");
println(trees); // Prints "lychee"
```
Array Functions

- The `expand()` function increases the size of an array.
  - It can expand to a specific size, or if no size is specified, the array’s length will be doubled.
  - If an array needs to have many additional elements, it’s faster to use `expand()` to double the size than to use `append()` to continually add one value.
Copying Arrays

- Array values cannot simply be copied with the assignment operator.
  - One way to copy from one array to another is to write a `for` loop.
  - The `arraycopy()` function is the most efficient way to copy from one array to another.
    - The data is copied from the first array to the second array, given as parameters.
    - Both arrays must be the same length.
String[] north = { "OH", "IN", "MI" };  
String[] south = { "GA", "FL", "NC" };  
arraycopy(north, south); // Copy from north array to south array  
println("south = ");  
println(south);       // Prints "OH IN MI"

String[] east = { "MA", "NY", "RI" };  
String[] west = new String[east.length]; // Create a new array  
arraycopy(east, west); // Copy from east array to west array  
println("west = ");  
println(west);       // Prints "MA NY RI"
Passing Arrays to Functions

- New functions can be written to perform operations on arrays, but arrays behave differently than data types such as `int` and `char`.
- When an array is used as a parameter to a function, the address (location in memory) of the array is transferred into the function instead of the actual data. No new array is created, and changes made within the function affect the array used as the parameter.
float[] data = {19.0, 40.0, 75.0, 76.0, 90.0};

void setup() {
    halve(data);
    println(data[0]); // Prints "9.5"
    println(data[1]); // Prints "20.0"
    println(data[2]); // Prints "37.5"
    println(data[3]); // Prints "38.0"
    println(data[4]); // Prints "45.0"
}

void halve(float[] d) {
    for (int i = 0; i < d.length; i++) {
        d[i] = d[i] / 2.0; // For each array element, divide the value by 2
    }
}
float[] data = {19.0, 40.0, 75.0, 76.0, 90.0};
float[] halfData;

void setup() {
    halfData = halve(data); // Run the halve() function
    println(data[0] + ', ' + halfData[0]); // Prints "19.0, 9.5"
    println(data[1] + ', ' + halfData[1]); // Prints "40.0, 20.0"
    println(data[2] + ', ' + halfData[2]); // Prints "75.0, 37.5"
    println(data[3] + ', ' + halfData[3]); // Prints "76.0, 38.0"
    println(data[4] + ', ' + halfData[4]); // Prints "90.0, 45.0"
}

float[] halve(float[] d) {
    float[] numbers = new float[d.length]; // Create a new array
    arraycopy(d, numbers); // Copy the contents
    for (int i = 0; i < numbers.length; i++) { // For each element,
        numbers[i] = numbers[i] / 2; // divide the value by 2
    }
    return numbers; // Return the new array
}
Two-Dimensional Arrays

- Data can also be stored and retrieved from arrays with more than one dimension.

- A 2D array is essentially a list of 1D arrays. It must be declared, then created, and then the values can be assigned just as in a 1D array.

- The data points for the star can be put into a 2D array instead of two 1D arrays:

<table>
<thead>
<tr>
<th>points</th>
<th>[0]</th>
<th>50</th>
<th>61</th>
<th>83</th>
<th>69</th>
<th>71</th>
<th>50</th>
<th>29</th>
<th>31</th>
<th>17</th>
<th>39</th>
</tr>
</thead>
<tbody>
<tr>
<td>[1]</td>
<td>18</td>
<td>37</td>
<td>43</td>
<td>60</td>
<td>82</td>
<td>73</td>
<td>82</td>
<td>60</td>
<td>43</td>
<td>37</td>
<td></td>
</tr>
</tbody>
</table>

<table>
<thead>
<tr>
<th>[0]</th>
<th>[1]</th>
<th>[2]</th>
<th>[3]</th>
<th>[4]</th>
<th>[5]</th>
<th>[6]</th>
<th>[7]</th>
<th>[8]</th>
<th>[9]</th>
</tr>
</thead>
</table>
```java
int[][] points = { {50, 18}, {61, 37}, {83, 43}, {69, 60}, 
{71, 82}, {50, 73}, {29, 82}, {31, 60}, 
{17, 43}, {39, 37} }; 

void setup() {
    size(100, 100);
    fill(0);
    smooth();
}

void draw() {
    background(204);
    translate(mouseX - 50, mouseY - 50);
    beginShape();
    for (int i = 0; i < points.length; i++) {
        vertex(points[i][0], points[i][1]);
    }
    endShape();
}
```
Using Arrays for Animation

- Arrays can be used to store and draw a sequence of images to produce an animation.
- The frames of the animation should be loaded in the `setup()` function and the `draw()` function should choose one image to draw on each frame.
int numFrames = 12; // The number of animation frames
PImage[] images = new PImage[numFrames]; // Image array

void setup() {
    size(100, 100);
    frameRate(30); // Maximum 30 frames per second

    // Automate the image loading procedure. Numbers less than 100
    // need an extra zero added to fit the names of the files.
    for (int i = 0; i < images.length; i++) {
        // Construct the name of the image to load
        String imageName = "ani-" + nf(i, 3) + ".gif";
        images[i] = loadImage(imageName);
    }
}

void draw() {
    // Calculate the frame to display, use % to cycle through frames
    int frame = frameCount % numFrames;
    image(images[frame], 0, 0);
}
int numFrames = 12; // The number of animation frames
int topFrame = 0; // The top frame to display
int bottomFrame = 0; // The bottom frame to display
PImage[] images = new PImage[numFrames];
int lastTime = 0;

void setup() {
  size(100, 100);
  frameRate(30);
  for (int i = 0; i < images.length; i++) {
    images[i] = loadImage("ani-" + nf(i, 3) + ".gif");
  }
}

void draw() {
  topFrame = (topFrame + 1) % numFrames;
  image(images[topFrame], 0, 0);
  if ((millis() - lastTime) > 500) {
    bottomFrame = (bottomFrame + 1) % numFrames;
    lastTime = millis();
  }
  image(images[bottomFrame], 0, 50);
}
void setup() {
    size(100, 100);
    img = loadImage("PT-Shifty-0020.gif");
}

void draw() {
    background(204);
    x += 0.5;
    if (x > width) {
        x = -width;
    }
    image(img, x, 0);
}
PImage img;
float angle;

void setup() {
  size(100, 100);
  img = loadImage("PT-Shifty-0023.gif");
}

draw() {
  background(204);
  angle += 0.01;
  translate(50, 50);
  rotate(angle);
  image(img, -100, -100);
}
Pixels
Getting Pixel Values

- The `get()` function can read the color of any pixel in the display window.
- It can also grab the whole display window or a section of it. There are three versions of this function, one for each use.

```plaintext
// Returns a copy of display window as a PImage
get()
// Returns color value of a single pixel
get(x, y)
// Returns a rectangular area as a PImage
get(x, y, width, height)
```
Getting the Display Window

strokeWeight(8);
line(0, 0, width, height);
line(0, height, width, 0);
PImage cross = get(); // Get the entire window
image(cross, 0, 50); // Draw the image in a new position
Getting a Pixel Rectangle

```
strokeWeight(8);
line(0, 0, width, height);
line(0, height, width, 0);
PImage slice = get(0, 0, 20, 100); // Get window section
set(18, 0, slice);
set(50, 0, slice);
```
Getting Single Pixel Values

PImage trees;
trees = loadImage("topanga.jpg");
noStroke();
image(trees, 0, 0);
color c = get(20, 30); // Get color at (20, 30)
fill(c);
rect(20, 30, 40, 40);
PImage trees;
int y = 0;

void setup() {
  size(100, 100);
  trees = loadImage("topangaCrop.jpg");
}

void draw() {
  image(trees, 0, 0);
  y = frameCount % height;
  for (int i = 0; i < 49; i++) {
    color c = get(i, y);
    stroke(c);
    line(i + 50, 0, i + 50, 100);
  }
  stroke(255);
  line(0, y, 49, y);
}
Getting Pixels From an Image

- Every PImage variable has its own `get()` to grab pixels from the image.
- This allows pixels to be grabbed from an image independently of the pixels in the display window.
- Because a PImage is an object, the `get()` function is accessed with the name of the image and the dot operator.
Getting Pixels From an Image

```java
PImage trees;
trees = loadImage("topanga.jpg");
stroke(255);
strokeWeight(12);
image(trees, 0, 0);
line(0, 0, width, height);
line(0, height, width, 0);
PImage treesCrop = trees.get(20, 20, 60, 60);
image(treesCrop, 20, 20);
```
Writing Pixel Values

- The pixels in Processing’s display window can be written directly with the `set()` function.

- There are two versions of this function, each with three parameters.

```plaintext
// Set the color of a single pixel
set(x, y, color)

// Set the pixel colors of a rectangle using a PImage
set(x, y, image)
```
Writing Single Pixels

```java
for (int i = 0; i < 55; i++) {
    for (int j = 0; j < 55; j++) {
        color c = color((i + j) * 1.8);
        set(30 + i, 20 + j, c);
    }
}
```
Using `set()` to Draw Images

- The `set()` function can write an image to the display window at any location.
- Using `set()` to draw an image is faster than using the `image()` function because the pixels are copied directly.
- However, images drawn with `set()` cannot be resized or tinted, and they are not affected by the transformation functions.
Using `set()` to Draw Images

```java
PImage trees;

void setup() {
    size(100, 100);
    trees = loadImage("topangaCrop.jpg");
}

void draw() {
    int x = frameCount % 50;
    set(x, 0, trees);
}
```
Writing Pixels in Images

- Every PImage variable has its own set() function to write pixels directly to the image.
  - This allows pixels to be written to an image independently of the pixels in the display window.
- Because a PImage is an object, the set() function is run with the name of the image and the dot operator.
Writing Pixels to Images

PImage trees;

void setup() {
    trees = loadImage("topangaCrop.jpg");
}

void draw() {
    background(0);
    int x = frameCount % trees.width;
    int y = frameCount % trees.height;
    color white = color(255);
    trees.set(x, y, white);
    trees.updatePixels();
    image(trees, 20, 0);
}
Lab Exercises
Lab Exercises

‣ Create an array to store the y-coordinates of a sequence of shapes.
  ‣ Draw each shape inside a for loop and use the values from the array to set the y-coordinate.

‣ Write a function to multiply two arrays together & return the result as a new array.
  ‣ Print the results to the console.
Lab Exercises

› Use a 2D array to store the coordinates for a shape of your own invention.
  › Use a for structure to draw the shape to the display window.

› Load a sequence of related images into an array and use them to create an animation.
  › Modify the program to present each frame of animation in a random sequence.
Lab Exercises

› Load an image and use `get()` to create a collage by overlaying different sections of the same image.

› Draw a shape in the display window. Copy a section of the window to another by using `get()` and `set()` within a `for` loop.